

```

1 ; Name      : simple.asm
2 ;
3 ; Build     : aclocal && autoconf && automake --add-missing --foreign
4 ;           mkdir build
5 ;           cd build
6 ;           ../configure
7 ;           make
8 ;
9 ; Description : a simple window with the basic functionalities and a title
10 ;
11 ; Source     : http://zetcode.com/gui/gtk2/firstprograms/
12
13 bits 64
14
15 [list -]
16     %define    GTK_WINDOW_TOPLEVEL    0
17     %define    GTK_WIN_POS_CENTER    1
18     %define    FALSE                0
19     %define    TRUE                 1
20     extern     gtk_init
21     extern     gtk_main
22     extern     gtk_main_quit
23     extern     gtk_widget_show
24     extern     gtk_window_new
25     extern     gtk_window_set_title
26     extern     g_signal_connect_data
27     extern     exit
28 [list +]
29
30 section .rodata
31     szTitle:    db    "simple",0
32     szDestroy:  db    "destroy",0
33
34 section .data
35     window:     dq    0
36
37 section .text
38 global _start
39
40 _start:
41     ;init gtk
42     xor     rdi,rdi
43     xor     rsi,rsi
44     call    gtk_init
45     ;if initialization doesn't succeed then the application
46     ;terminates. If that is unwanted behaviour then we need
47     ;to use gtk_init_check instead.
48     ;https://developer.gnome.org/gtk2/stable/gtk2-General.html#gtk-init
49     ;create a new window
50     mov     rdi,GTK_WINDOW_TOPLEVEL
51     call    gtk_window_new
52     mov     qword[window],rax
53     ;set the title
54     mov     rdi,rax
55     mov     rsi,szTitle
56     call    gtk_window_set_title
57     ;connect the destroy signal to gtk_main_quit event handler
58     xor     r9d,r9d                ; combination of GConnectFlags
59     xor     r8d,r8d                ; a GClosureNotify for data
60     xor     rcx,rcx                ; pointer to the data to pass
61     mov     rdx,gtk_main_quit      ; pointer to the handler
62     mov     rsi,szDestroy          ; pointer to the signal
63     mov     rdi,qword[window]      ; pointer to the widget instance
64     ;C programs uses g_signal_connect, this is not a library
65     ;function.
66     call    g_signal_connect_data
67     ;show the window
68     mov     rdi,qword[window]
69     call    gtk_widget_show
70     ;go into applications main loop
71     call    gtk_main
72 .exit:

```

```
73 ;exit program
74 xor    rdi,rdi
75 call   exit
76
```